
HEIZKURVENBERECHNUNG MIT DER C-CONTROL II 
Entspricht in etwa einer Viessmann Heizkurve (Trimatik) 
 
Vorlage: André Helbig / Geändert: Malte Alpers 
 
 
function getSoll(int Ta, int K, int P) returns int 
{ 
long TSoll; 
if Ta > 200 return 0; // Ta-Max. in 1/10°C (z.B. 20°C = 200) 
if Ta < -200 Ta=-200; // Ta-Min. in 1/10°C (z.B. -20°C = -200) 
TSoll=((Ta*-10)+4200); 
Ta=Ta/2; 
TSoll=(((TSoll+((Ta*Ta)/15)*-1+2000/3)-2200)*K)/1000+200+P; 
if TSoll > 900 return 900; // TSoll-Max. in 1/10°C (z.B. 90°C = 900) 
// if TSoll < 0 return 0; /* TSoll-Min. in 1/10°C (z.B. 25°C = 250), 
wenn aktiviert, Rückgabewert Ta-Max. entsprechend TSoll-Min. setzen! */ 
return TSoll; 
} 
// Tsoll : Solltemperatur (Vorlauf) in 1/10°C (z.B. 45,3°C = 453) 
// Ta : Außentemperatur in 1/10°C (z.B. 3,5°C = 35) 
// K : Steigung der Heizkurve in 1/100 ! (z.B. 0,8 = 80 ; 1,2 = 120) 
// P : Parallelverschiebung in 1/10 K (z.B. 1,4 K = 14) 
 



 
/*********************************************************************/ 
/* Testprogramm zur Ausgabe der berechneten Sollwerte der Heizkurve  */ 
/*********************************************************************/ 
/* - Auswertung über Hyperterminal in Textdatei speichern.           */ 
/*   Die Datei kann dann in einer Tabellenkalkulation geöffnet       */ 
/*   werden (mit Semikolon getrennte Werte)                          */ 
/* - Ausgabe: 'T' in 1/10°C  / 'K' in 1/100 !                        */ 
/*********************************************************************/ 
 
 
// hier 'function getSoll' einfügen 
 
 
thread main 
{ 
  int soll,c,d,P; 
  byte n,i; 
  string s; 
  hwcom.init(); 
  hwcom.setspeed(8); 
  sleep 100; 
  hwcom.clr(); 
   
  P=0; // parallel in 1/10 K (z.B 5K = 50) 
  s="P: "; 
  str.putint(s,P); 
  hwcom.print(s); 
  hwcom.ret(); 
  hwcom.ret(); 
   
  for n=0...12 // Anzahl der K-Werte 
  { 
    c=20+(20*n); // unterer K-Wert und step 
    s="K: "; 
    str.putint(s,c); 
    hwcom.print(s); 
    hwcom.ret(); 
    for i=0...12 // Anzahl der Temperaturwerte 
    { 
      d=-300+(50*i); // untere Temperatur und step 
      soll=getSoll(d, c, P); 
      s=""; 
      str.putint(s,d); 
      s=s+";"; 
      str.putint(s,soll); 
      hwcom.print(s); 
      hwcom.ret(); 
    } 
    hwcom.ret(); 
    hwcom.ret(); 
  } 
  halt; 
} 


